Computer Studies of Swirling Particle Fluids and the Evolution of Planetary-Type Bodies

by

Donald Greenspan

and

John Collier

Computer Sciences Technical Report #299

May 1977

Computer Studies of Swirling Particle Fluids and the Evolution of Planetary-Type Bodies

by

Donald Greenspan and John Collier

Computer Sciences Department
University of Wisconsin, Madison

Abstract

In this paper a particle-type model is applied to the study of evolving swirling fluids. Of basic importance is a natural, self-reorganizing property of the system. High-speed digital computation is essential for the resolution of associated initial-value problems. Examples are described which incorporate local, molecular-type forces and long-range, gravitational forces. Moon-like and planetary-like bodies thus evolve in a natural way. Extensive computer examples are given.

1. Introduction

In this paper we will initiate a new type of modeling for swirling fluid phenomena. Both local, molecular type, forces and long-range, gravitational, forces will be included in a fashion which is qualitatively similar to the way in which they actually occur. Self-gravitational interaction will yield a natural, self reorganization property in which heavier particles relocate centrally. Molecular type interactions will yield conservation of mass, will provide a mechanism for generating various pressure distributions, and will also contribute to the system's self-reorganization property. Much of the interest at present will focus on the process of solidification and the natural evolution of planetary type bodies.

2. Basic Mathematical Definitions and Formulas

For positive time step Δt , let $t_k = k\Delta t,\ k = 0,1,2 \ldots$. For $i = 1,2,3,\ldots,N$, let particle P_i have mass m_i and at time t_k let P_i be located at $\vec{r}_{i,k} = (x_{i,k}, y_{i,k})$, have velocity $\vec{v}_{i,k} = (v_{i,k,x}, v_{i,k,y})$, and have acceleration $\vec{a}_{i,k} = (a_{i,k,x}, a_{i,k,y})$. Let position, velocity and acceleration be related by the "leap-frog" formulas ([2], p. 107):

$$(2.1) \qquad \vec{v}_{i,\frac{1}{2}} = \vec{v}_{i,0} + \frac{\Delta t}{2}\,\vec{a}_{i,0}$$

$$(2.2) \qquad \vec{v}_{i,k+\frac{1}{2}} = \vec{v}_{i,k-\frac{1}{2}} + (\Delta t)\,\vec{a}_{i,k} \qquad ,\ k = 1,2,\ldots .$$

$$(2.3) \qquad \vec{r}_{i,k+1} = \vec{r}_{i,k} + (\Delta t)\,\vec{v}_{i,k+\frac{1}{2}} \qquad ,\ k = 0,1,2,\ldots .$$

If $\vec{F}_{i,k}$ is the force acting on P_i at time t_k , where $\vec{F}_{i,k} = (F_{i,k,x}, F_{i,k,y})$, then we assume that force and acceleration are related by

$$(2.4) \qquad \vec{F}_{i,k} = m_i\,\vec{a}_{i,k} .$$

Once an exact structure is given to $\vec{F}_{i,k}$, the motion of each particle will be determined recursively and explicitly by (2.1)-(2.4) from prescribed initial data.

In the present paper, we will want $\vec{F}_{i,k}$ to incorporate both local, molecular type components and long-range, gravitational type components, and this will be implemented as follows. At time t_k, let $r_{ij,k}$ be the distance between P_i and P_j. Let G_{ij} (coefficient of molecular-type attraction), H_{ij} (coefficient of molecular-type repulsion), β_{ij} (exponent of molecular-type attraction), α_{ij} (exponent of molecular-type repulsion) and G^*_{ij} (gravitational constant) be constants determined by P_i and P_j, subject to the constraints (see [5]): $G_{ij}\geq 0$, $H_{ij}\geq 0$, $\alpha_{ij}\geq\beta_{ij}\geq 2$, $G^*_{ij}\geq 0$. Then the force $(\bar{F}_{i,k,x},\bar{F}_{i,k,y})$ exerted on P_i by P_j is defined by

$$(2.5)\qquad \bar{F}_{i,k,x} = \left[\frac{-G_{ij}}{r_{ij,k}^{\beta_{ij}}} + \frac{H_{ij}}{r_{ij,k}^{\alpha_{ij}}} - \frac{G^*_{ij}}{r_{ij,k}^{2}}\right]\frac{(m_i m_j)(x_{i,k}-x_{j,k})}{r_{ij,k}}$$

$$(2.6)\qquad \bar{F}_{i,k,y} = \left[\frac{-G_{ij}}{r_{ij,k}^{\beta_{ij}}} + \frac{H_{ij}}{r_{ij,k}^{\alpha_{ij}}} - \frac{G^*_{ij}}{r_{ij,k}^{2}}\right]\frac{(m_i m_j)(y_{i,k}-y_{j,k})}{r_{ij,k}} .$$

The total force $(F_{i,k,x},F_{i,k,y})$ on P_i due to all the other $(N-1)$ particles is given by

$$(2.7)\qquad F_{i,k,x} = \sum_{\substack{j=1\\ j\neq i}}^{N} \bar{F}_{i,k,x} ;\quad F_{i,k,y} = \sum_{\substack{j=1\\ j\neq i}}^{N} \bar{F}_{i,k,y} .$$

The formulation (2.1)-(2.7) is explicit and economical though non-conservative. Conservation of energy and momenta can be achieved [2], but only through an implicit, less economical approach. Throughout, the time step to be used in (2.1)-(2.3) will always be $\Delta t = 10^{-4}$ and a comprehensive FORTRAN program for implementation of (2.1)-(2.7) is given in the Appendix of [4]. It should be noted, in addition, that the way in which (2.1)-(2.7) will be applied lends itself directly to parallel computation also.

3. Basic Physical Assumptions and Definitions

We will consider a system of 239 particles, so that N = 239. This parameter was determined solely by economic constraints. Next, we fix the parameters $\alpha_{ij} \equiv 6$, $\beta_{ij} \equiv 4$, which were shown to be viable in previous computations [3] with relatively smaller particle sets. Unless indicated otherwise, initial particle positions will always be those shown in Figure 1 and listed precisely in the x_0, y_0 columns of Table I. These positions were generated in such a fashion that the configuration is approximately circular and is, for zero initial velocities and constant masses, relatively stable [2].

The entire configuration will be set into counterclockwise rotation as follows. In terms of the angular velocity parameter $\dot{\theta}$ and a perturbation parameter ε, let

$$(3.1) \qquad v_{i,0,x} = \pm |\dot{\theta} y_{i,0}| \pm \frac{m_i}{2000}\varepsilon, \quad v_{i,0,y} = \pm |\dot{\theta} x_{i,0}| \pm \frac{m_i}{2000}\varepsilon,$$

where the choices of the signs will be made as follows. Choose the signs before the absolute value terms in (3.1) by setting $\varepsilon = 0$ and using the rule

$$(x_{i,0},y_{i,0}) \in \text{Quadrant I} \Rightarrow v_{i,0,x} \le 0,\ v_{i,0,y} \ge 0$$
$$(x_{i,0},y_{i,0}) \in \text{Quadrant II} \Rightarrow v_{i,0,x} \le 0,\ v_{i,0,y} \le 0$$
$$(x_{i,0},y_{i,0}) \in \text{Quadrant III} \Rightarrow v_{i,0,x} \ge 0,\ v_{i,0,y} \le 0$$
$$(x_{i,0},y_{i,0}) \in \text{Quadrant IV} \Rightarrow v_{i,0,x} \ge 0,\ v_{i,0,y} \ge 0 .$$

FIGURE I - INITIAL POSITIONS



TABLE I - INITIAL CONDITIONS

Mass	x_0	y_0	$v_{x,0}$	$v_{y,0}$
10000.000	-2.013	3.447	-63.700	-58.093
10000.000	.487	2.590	-60.548	-48.656
8000.000	-4.489	-2.614	50.506	-57.819
8000.000	1.496	.874	-36.596	34.207
8000.000	-2.503	2.595	-29.911	-50.305
8000.000	-2.486	-6.071	64.515	-49.925
6000.000	2.005	-5.198	49.930	22.043
6000.000	-.991	-1.736	36.643	-34.130
6000.000	5.497	-.869	-33.072	8.022
6000.000	-5.494	.858	26.435	-52.036
6000.000	-1.002	1.727	-23.062	-32.590
6000.000	.989	5.197	-8.967	-26.022
6000.000	-1.991	-3.469	44.047	-38.052
6000.000	4.504	-2.599	39.992	11.172
6000.000	3.499	4.302	-11.928	16.192
6000.000	2.002	-1.726	23.000	36.942
6000.000	-3.004	3.448	-42.176	-17.779
6000.000	4.991	1.742	-36.756	50.254
4000.000	4.510	-6.050	44.031	38.588
4000.000	-4.990	-5.202	41.108	.630
4000.000	-3.499	2.601	10.161	6.379
4000.000	-2.487	-7.799	51.533	11.061
4000.000	6.501	2.613	8.545	45.730
4000.000	-1.987	-6.929	48.029	12.783
4000.000	-.498	-4.334	-2.860	-21.963
4000.000	4.996	.002	-19.979	.281
4000.000	-1.498	-4.337	-2.833	-26.232
4000.000	-5.507	2.590	-30.227	-42.401
4000.000	-1.513	7.793	-51.057	-26.719
4000.000	-3.001	-1.737	-13.200	-32.448
4000.000	6.010	-5.198	40.072	4.550
4000.000	.510	-6.064	44.571	-17.775
4000.000	-1.990	-5.207	40.949	-27.638
4000.000	5.007	-3.458	33.898	-.198
4000.000	1.007	-3.464	33.939	-16.214
4000.000	3.007	-1.731	27.145	-8.222
4000.000	-6.496	-.883	24.141	-45.653
4000.000	-2.004	.001	-19.751	11.913
4000.000	4.496	.880	-23.345	37.608
4000.000	1.493	2.607	-30.421	25.221
4000.000	4.993	3.470	-34.039	39.965
4000.000	-4.510	4.328	-36.988	1.559
4000.000	3.990	5.210	-41.064	35.175
4000.000	-2.013	6.931	-47.372	11.412
2000.000	-2.507	7.799	-20.876	-.619
2000.000	3.011	-6.925	37.353	22.904
2000.000	1.508	-6.055	34.193	15.821
2000.000	-4.492	-6.064	34.856	-7.500
2000.000	.008	-5.198	30.864	10.301
2000.000	-.992	-5.198	30.465	5.822
2000.000	.508	-4.328	27.603	11.559
2000.000	1.508	-4.325	27.164	15.831
2000.000	7.005	-3.449	22.975	38.066
2000.000	-5.995	-3.464	24.863	-13.616

TABLE I - CONTINUED

Mass	x_0	y_0	$v_{x,0}$	$v_{y,0}$
2000.000	3.505	-2.595	20.310	24.138
2000.000	-3.495	-2.601	20.675	-4.072
2000.000	6.005	-1.722	16.436	33.648
2000.000	.005	-1.728	17.189	9.776
2000.000	-5.995	-1.734	17.642	-14.478
2000.000	3.502	-.865	13.802	23.993
2000.000	-2.498	-.871	13.871	-.112
2000.000	6.002	.008	9.528	34.214
2000.000	3.002	.005	9.612	22.053
2000.000	-5.998	-.004	10.134	-14.016
2000.000	-1.498	.869	6.335	4.096
2000.000	-.498	.872	6.963	8.070
2000.000	-6.498	.863	7.098	-16.223
2000.000	-2.498	.869	6.860	.649
2000.000	.999	1.732	2.741	14.224
2000.000	-5.001	1.726	3.227	-9.785
2000.000	-4.001	1.726	3.055	-5.651
2000.000	-6.501	2.593	.323	-16.450
2000.000	-.001	3.462	-4.250	10.021
2000.000	.496	4.332	-7.248	11.301
2000.000	-1.504	4.329	-7.790	3.488
2000.000	-6.504	4.323	-6.603	-16.562
2000.000	1.992	5.201	-30.436	-2.226
2000.000	-6.008	5.192	-30.084	-34.604
2000.000	-.508	6.058	-34.086	-11.939
2000.000	1.989	6.931	-38.189	-2.813
2000.000	1.489	7.801	-41.338	-4.720
2000.000	2.507	-7.799	20.868	.614
2000.000	-.993	-6.932	17.339	-13.160
2000.000	2.504	-6.059	13.969	.100
2000.000	-3.496	-6.065	14.934	-23.588
2000.000	1.004	-5.202	10.675	-5.577
2000.000	-2.996	-5.205	11.046	-21.623
2000.000	2.504	-4.329	7.105	.065
2000.000	6.001	-3.456	3.199	14.164
2000.000	.001	-3.462	3.390	-9.927
2000.000	-6.999	-3.471	4.524	-37.853
2000.000	-2.499	-2.605	.397	-19.737
2000.000	7.001	-1.723	-4.134	18.136
2000.000	-4.999	-1.738	-3.122	-30.063
2000.000	4.498	-.866	-6.699	7.907
2000.000	-1.502	-.875	-6.675	-15.910
2000.000	6.998	.007	-11.135	17.783
2000.000	3.998	.004	-9.577	5.837
2000.000	-7.002	-.011	-8.929	-37.800
2000.000	.498	.868	-13.788	-7.132
2000.000	-4.502	.862	-13.455	-27.916
2000.000	-3.502	.865	-13.429	-23.805
2000.000	-.005	1.728	-16.122	-10.123
2000.000	-6.005	1.722	-16.781	-33.875
2000.000	7.495	2.607	-21.049	19.735
2000.000	-4.505	2.592	-19.992	-27.953
2000.000	.995	3.458	-23.847	-5.399
2000.000	-1.005	3.458	-23.940	-14.888

TABLE I - CONTINUED

Mass	x_0	y_0	$v_{x,0}$	$v_{y,0}$
2000.000	4.492	4.334	-27.368	7.903
2000.000	5.492	4.334	-28.057	11.571
2000.000	1.492	4.331	-27.334	-4.517
2000.000	-.508	4.328	-27.502	-12.539
2000.000	-5.008	5.192	-30.470	-30.635
2000.000	-1.508	6.055	-34.261	-15.724
2000.000	2.989	6.931	-37.991	1.291
2000.000	-1.011	6.928	-37.724	-14.467
2000.000	1.507	-7.799	20.994	-3.165
2000.000	-1.493	-7.805	21.211	-15.021
2000.000	.007	-6.932	17.634	-9.508
2000.000	3.504	-6.059	13.461	4.619
2000.000	3.004	-5.199	10.844	2.132
2000.000	-3.996	-5.208	11.604	-25.662
2000.000	3.504	-4.329	7.267	3.980
2000.000	-2.496	-4.335	7.436	-20.447
2000.000	4.001	-3.456	3.750	5.980
2000.000	-.999	-3.462	3.939	-13.799
2000.000	-4.999	-3.468	3.800	-30.306
2000.000	5.501	-2.596	.415	12.575
2000.000	-1.499	-2.605	.369	-15.970
2000.000	-7.495	-2.611	21.132	-39.606
2000.000	1.005	-1.732	16.755	-6.301
2000.000	-3.995	-1.738	16.866	-26.138
2000.000	6.502	-.863	12.822	16.154
2000.000	-.498	-.872	13.320	-11.742
2000.000	.502	-.872	13.561	-7.918
2000.000	-7.498	-.881	14.141	-39.959
2000.000	-5.498	-.878	13.698	-31.834
2000.000	2.002	.001	10.032	-2.358
2000.000	-3.998	-.008	9.796	-25.846
2000.000	3.502	.871	6.472	4.005
2000.000	5.502	.874	6.538	11.540
2000.000	5.999	1.734	2.619	14.059
2000.000	1.999	1.731	2.607	-1.494
2000.000	-7.001	1.719	4.261	-38.051
2000.000	5.499	2.604	-.471	11.759
2000.000	-1.501	2.595	-.292	-16.074
2000.000	-7.501	2.589	.356	-40.266
2000.000	1.999	3.461	-4.224	-1.783
2000.000	2.496	4.331	-7.565	.214
2000.000	-2.504	4.325	-7.488	-20.654
2000.000	2.996	5.201	-11.085	1.652
2000.000	-4.004	5.192	-10.465	-26.695
2000.000	.496	6.058	-14.298	-7.829
2000.000	-3.504	6.055	-13.721	-24.397
2000.000	-3.007	6.925	-17.259	-22.515
2000.000	.493	7.798	-21.302	-8.712
2000.000	-.489	-7.802	41.037	-11.088
2000.000	1.011	-6.932	37.844	-5.258
2000.000	-1.492	-6.065	34.425	-15.952
2000.000	4.008	-5.196	30.394	6.678
2000.000	-5.992	-5.208	31.333	-33.508
2000.000	5.508	-4.326	26.765	12.535

TABLE I - CONTINUED

Mass	x_0	y_0	$v_{x,0}$	$v_{y,0}$
2000.000	-3.492	-4.335	27.649	-23.920
2000.000	-6.492	-4.341	27.976	-35.575
2000.000	2.005	-3.459	23.911	-2.091
2000.000	-3.995	-3.468	24.091	-26.184
2000.000	6.505	-2.593	19.268	16.339
2000.000	-.495	-2.602	20.272	-12.374
2000.000	.505	-2.602	20.206	-7.595
2000.000	-6.495	-2.611	21.409	-35.547
2000.000	4.005	-1.726	17.254	6.058
2000.000	7.502	-.863	12.538	20.041
2000.000	1.502	-.869	13.649	-3.339
2000.000	-4.498	-.878	13.329	-28.419
2000.000	-.998	-.002	9.984	-14.004
2000.000	-4.998	-.008	9.795	-30.092
2000.000	2.502	.871	6.578	.041
2000.000	6.999	1.737	2.216	18.113
2000.000	-7.498	.859	7.528	-40.161
2000.000	2.999	1.731	2.686	1.761
2000.000	-3.001	1.725	3.732	-21.165
2000.000	3.499	2.601	-.030	4.079
2000.000	2.495	2.605	-20.882	19.700
2000.000	-.505	2.602	-21.368	8.316
2000.000	5.995	3.468	-24.354	33.715
2000.000	2.995	3.465	-23.897	21.952
2000.000	-4.005	3.456	-23.030	-6.161
2000.000	-7.005	3.453	-23.035	-18.228
2000.000	-6.005	3.456	-23.254	-14.033
2000.000	-3.508	4.329	-26.368	-4.922
2000.000	4.992	5.208	-31.370	28.990
2000.000	-3.008	5.199	-30.563	-2.974
2000.000	-2.008	5.199	-30.894	1.100
2000.000	3.492	6.065	-34.676	23.371
2000.000	1.492	6.065	-34.222	16.098
2000.000	-4.508	6.056	-33.902	-8.755
2000.000	.989	6.932	-37.717	13.087
2000.000	-.511	7.802	-41.393	7.170
2000.000	.507	-7.798	21.205	12.933
2000.000	2.007	-6.925	17.199	19.126
2000.000	-2.993	-6.931	18.161	-1.283
2000.000	-.496	-6.058	13.606	8.178
2000.000	5.004	-5.192	10.512	30.881
2000.000	6.504	-4.319	6.586	36.567
2000.000	4.504	-4.322	7.385	28.207
2000.000	-5.496	-4.334	7.942	-11.196
2000.000	-4.496	-4.334	7.393	-7.217
2000.000	3.001	-3.455	3.874	22.206
2000.000	-2.999	-3.461	3.481	-1.801
2000.000	7.501	-2.589	-.254	40.267
2000.000	1.501	-2.595	.616	16.211
2000.000	2.501	-2.595	.397	20.005
2000.000	-5.499	-2.604	.373	-11.792
2000.000	5.001	-1.722	-3.288	29.701
2000.000	-1.999	-1.731	-3.065	1.824
2000.000	-6.999	-1.737	-2.428	-18.136

TABLE I - CONCLUDED

Mass	x_0	y_0	$v_{x,0}$	$v_{y,0}$
2000.000	2.498	-.865	-6.705	20.480
2000.000	-3.502	-.871	-6.532	-4.177
2000.000	-.002	.002	-9.843	10.323
2000.000	.998	.002	-9.866	13.895
2000.000	-3.002	-.001	-9.997	-2.150
2000.000	6.498	.881	-14.662	35.882
2000.000	7.498	.881	-14.188	40.037
2000.000	3.995	1.738	-16.825	26.147
2000.000	-2.005	1.729	-17.136	3.032
2000.000	4.495	2.608	-20.408	27.820
2000.000	6.995	3.471	-24.640	37.876
2000.000	3.995	3.468	-23.657	25.962
2000.000	-5.005	3.456	-23.512	-9.969
2000.000	6.492	4.341	-28.043	35.246
2000.000	-5.508	4.326	-26.915	-12.609
2000.000	5.992	5.208	-31.539	33.386
2000.000	-1.008	5.202	-30.861	5.542
2000.000	-.008	5.202	-30.831	9.907
2000.000	4.492	6.068	-34.644	27.349
2000.000	2.492	6.065	-34.604	19.349
2000.000	-2.508	6.059	-33.919	-.054
2000.000	-.011	6.932	-37.845	9.354
2000.000	2.499	7.805	-41.533	19.388

As will be discussed later, the signs before the ε terms will be determined at random.

Still another parameter which will be important will be a distance parameter D which will determine whether the long range forces or the short range forces predominate. We will choose $D>1$ and will use the "switching" rule

$$(3.2) \qquad r_{ij,k} < D \Rightarrow G^*_{ij} \equiv 0$$

$$(3.3) \qquad r_{ij,k} \geq D \Rightarrow G_{ij} \equiv H_{ij} \equiv 0 \ .$$

Further, as in [3], we will fix $G_{ij} \equiv H_{ij} \equiv 5$ whenever (3.2) is valid. Thus, once D is assigned, G_{ij} and H_{ij} are completely determined, but G^*_{ij} is, as yet, still arbitrary whenever (3.3) is valid.

In the assignment of particle masses, let us first consider the chemical composition of Earth, as shown in Table II. The proportionate total mass for each element is shown in the right-most column. If one regroups all the elements in Table II as shown in Table III, then one observes that, in order, the total masses of groups I-V decrease, while the individual particle masses increase. It is this observation which is incorporated qualitatively into our model as follows. We will consider five groups, consisting of 2, 4, 12, 26, and 197 particles, respectively, with individual particle masses of 10000, 8000, 6000, 4000, and 2000 units, respectively. The actual assignment of a mass to each particle will be implemented later by a random process.

TABLE II - CHEMICAL COMPOSITION OF EARTH

Element	Relative number of atoms	Element's atomic weight	Product
Hydrogen	40000.0	1.008	40320
Helium	3100.0	4.003	12409
Carbon	3.5	12.01	42
Nitrogen	6.6	14.008	92
Oxygen	21.5	16.000	344
Neon	8.6	20.183	174
Sodium	0.04	22.997	1
Magnesium	0.91	24.31	22
Aluminum	0.09	26.98	2
Silicon	1.0	28.09	28
Phosphorous	0.01	30.07	0
Sulfur	0.37	32.06	12
Argon	0.15	39.948	6
Calcium	0.05	40.08	2
Iron	0.6	55.85	34
Nickel	0.03	58.71	2

TABLE III - REGROUPING OF EARTH'S ELEMENTS

Group	Elements	Total Products
I	Hydrogen, Helium	52729
II	Carbon through Oxygen	478
III	Neon through Aluminum	199
IV	Silicon through Calcium	48
V	Iron, Nickel	36

Next, while in rotation we will require a rule for determining the physical state of each particle. Such matters can be exceedingly complex [1] so that, for the present, we will use the following intuitive notions. Consider four particles P_1, P_2, P_3, P_4, each of the same mass, located at (0,1), (-0.87, 0.5), (0.87,0.5), (0,0), respectively. If each particle is assigned $\vec{0}$ velocity, then the three particles P_1, P_2, P_4 form a three particle bond [2], since, by (2.5)-(2.7), the local force on any one due to the other two is zero. Similarly, P_1, P_3, P_4 form a three particle bond. We will then call P_1, P_2, P_3 and P_4 solid particles. To explore the change of state to fluid particles, we now change only the initial velocity of P_1 to $(0,v)$, where $v>0$, and seek the smallest value of v for which P_1 relocates monotonically upward, in accordance with (2.1)-(2.7), so that the given bonds are broken and new ones formed by the triplet P_1, P_2, P_3 and by the triplet P_2, P_3, P_4. In this fashion, the speed v assigned to P_1 has enabled the particles to change their bonds easily, which we call a fluid state, and, in particular, a liquid state. For $D = 2.1$, so that (3.2) is valid, these values of v are given in Table IV in the "v-liquid" column for the different masses to be considered. (Of course, these results are also valid for any $D>2.1$). However, it will be more convenient to identify a particle as being a liquid particle by its temperature, which is defined as follows [2]. The instantaneous temperature $T'_{i,k}$ of P_i at t_k is defined by its kinetic energy, that is

$$(3.4) \qquad T'_{i,k} = \frac{1}{2} m_i v_{i,k}^2 .$$

TABLE IV - Change of State Velocities and Temperatures

m_i	v-liquid	v-gas	temp-liquid	temp-gas
10000	100	170	11370	30200
8000	90	160	8190	22600
6000	78	140	4640	12500
4000	65	110	1160	5300
2000	50	80	710	2130

Since these numbers can be relatively large due to the magnitudes m_i, the normalized instantaneous temperature $T^*_{i,k}$ of P_i at t_k is defined by

$$T^*_{i,k} = T'_{i,k}/10^4 .$$

The temperature $T_{i,k}$ of P_i at t_k is defined by [2]:

$$(3.5) \qquad T_{i,k} = \frac{1}{M} \sum_{j=k-M}^{k} T^*_{i,j} ,$$

where M is a positive integer, and where (3.5) is an average over M time steps, thus corresponding to the fact that temperature is a quantity which is measured over a finite, positive time period. Computations have shown that it is reasonable in the examples which follow to choose $M = 500$, so that we will use

$$(3.6) \qquad T_{i,k} = \frac{1}{500} \sum_{j=k-500}^{k} T^*_{i,j} .$$

The temperatures at which each particle changes state from solid to liquid are listed in the "temp-liquid" column of Table IV.

The critical velocities and temperatures of gas particles, as shown in Table IV, were determined more simply by considering only three particles P_1, P_2, P_3, located at (0,0.87), (-0.5,0), (0.5,0), respectively, with initial velocities (0,v), (0,0), (0,0), respectively, and by determining the positive parameter v for which $|P_1P_2| = |P_1P_3| > D$.

The results shown are for $D = 2.3$. Intuitively, when $|P_1P_2| > D$, all molecular-type forces are zero, by (2.3), so that P_1 moves "freely", which is characteristic of gas particles.

Note that "temperature", as defined above, is a phenomenon of a particles "local" velocity, that is, its velocity relative to neighboring particles. Thus, when a particle is rotating within a large system, the gross system velocities should have no effect on the particle's temperature and must be subtracted out before the temperature calculation is performed. The velocity of the centroid of the system and of the average angular velocity of the system are utilized for this purpose in the following way to determine the temperature of P_i at t_k as P_i rotates within the system. First, at time t_k , let the mass center of the system be $(\bar{x}_k, \bar{y}_k)$ and let the average linear velocity, $(\bar{v}_{x,k}, \bar{v}_{y,k})$, of the system be defined by

$$(3.6) \qquad \bar{v}_{x,k} = \frac{\sum(m_i v_{x,i,k})}{\sum m_i} , \quad \bar{v}_{y,k} = \frac{\sum(m_i v_{y,i,k})}{\sum m_i} ,$$

where the summations of (3.6) are taken over all particles of the system. Then P_i's position $(x^*_{i,k}, y^*_{i,k})$ and velocity $(v^*_{i,k,x}, v^*_{i,k,y})$ relative to the mass center are defined by

$$(3.7) \qquad x^*_{i,k} = x_{i,k} - \bar{x}_k , \quad y^*_{i,k} = y_{i,k} - \bar{y}_k$$

$$(3.8) \qquad v^*_{i,k,x} = v_{i,k,y} - \bar{v}_{x,k} , \quad v^*_{i,k,y} = v_{i,k,y} - \bar{v}_{y,k} .$$

Next, out of $v^*_{i,k,x}$ and $v^*_{i,k,y}$ we wish to take the angular rotation of the system, which is done as follows. Introduce the normal and tangent velocity components, $v^*_{i,k,n}$ and $v^*_{i,k,t}$, respectively, of P_i at t_k, by the usual formulas

(3.9) $$v^*_{i,k,n} = [v^*_{i,k,y}\, y^*_{i,k} + v^*_{i,k,x}\, x^*_{i,k}]/R_i$$

(3.10) $$v^*_{i,k,t} = [-v^*_{i,k,x}\, y^*_{i,k} + v^*_{i,k,y}\, x^*_{i,k}]/R_i \ ,$$

where

(3.11) $$R_i = [(x^*_{i,k})^2 + (y^*_{i,k})^2]^{\frac{1}{2}}$$

Since, in general, $\dot{\theta} = v_t/R$, we define the average angular velocity, $\bar{\dot{\theta}}$, of the system by

(3.12) $$\bar{\dot{\theta}} = \frac{\sum(\dot{\theta}_i m_i)}{\sum m_i} = \frac{\sum\left(m_i \dfrac{v^*_{i,k,t}}{R_i}\right)}{\sum m_i} \ ,$$

where the summations are taken over all the particles of the system. Finally, the speeds $v^2_{i,k}$, used in (3.4) to calculate the temperature by (3.6), are given by

(3.13) $$v^2_{i,k} = (v^*_{i,k,t} - \bar{\dot{\theta}} R_i)^2 + (v^*_{i,k,n})^2 \ .$$

As a last physical consideration, we will allow for radiation of heat into and out of the system. At any time, a particle P_i whose distance from the mass center is greater than d^* units ($d^* = 7$ unless otherwise indicated) is called an outer particle. An outer particle will be called a light-side particle, that is, it faces a sun, if $y_{i,k} \geq \bar{y}_k$. Otherwise, it is called a dark-side particle. Light-side particles will receive radiant heat while dark-side particles will emanate radiant heat. This is implemented as follows. Every k^* steps, outer particle velocities are reset by the rule

$$(3.14) \qquad \begin{aligned} &\text{light-side particle: } \vec{v}_{i,k} \to 1.001\, \vec{v}_{i,k} \\ &\text{dark-side particle: } \vec{v}_{i,k} \to 0.9955\, \vec{v}_{i,k} \,. \end{aligned}$$

In practice k^* will be fixed so that a dark-side particle will lose about 30% of its kinetic energy during its initial half-revolution as a dark side particle, which conforms with some estimates of radiant heat loss on the dark side of Earth. For $\dot{\theta} = 1$, $k^* = 200$, approximately, so that for $\dot{\theta} = 4$, $k^* = 50$, while for $\dot{\theta} = 5$, $k^* = 40$, which will be utilized in the examples which follow.

4. Examples

Before describing in detail the more interesting computer examples run, we will summarize the preliminary considerations and calculations related to the selection of the parameters $\dot{\theta}$, ε, D, k^* and G^*_{ij} .

Initially, the parameter choices were $\dot{\theta} = 1$, $D = 2.1$, $\varepsilon = 0$, $k^* = 200$ and $G^*_{ij} \equiv 0$, with a mass distribution generated at random. The resulting motion of the system was that of a solid body which, over 80000 times steps, showed small expansions on the light-side and small contractions on the dark-side. However, before parameter modifications could be introduced which would result in a rotating fluid body, economic constraints forced a change in $\dot{\theta}$. With $\dot{\theta} = 1$, one complete revolution of the system required,approximately, 50000 time steps, which was considered to be exorbitant. Studies were then made with $\dot{\theta} = 4, 5, 10, 20, 40$ and with,respectively, $k^* = 50, 40, 20, 10, 5$. The choices $\dot{\theta} = 10, 20, 40$ were relatively unstable over long periods, while $\dot{\theta} = 4, 5$ were reasonable. To affect a change of state from solid to fluid, ε was varied next. With $\dot{\theta} = 4, 5$, computer examples with $\varepsilon = 1, 5, 10, 20, 40$ showed that $\varepsilon = 10, 20$ yielded fluid type motions, while $\varepsilon = 40$ yielded instability. However, $\varepsilon = 10, 20$ showed exceptionally slow motions of the heavier particles toward the core, so that, next, with $\dot{\theta} = 4, 5$ and $\varepsilon = 10, 20$, further examples were run with $D = 2.3, 2.5, 3.1$. The resulting motions of the heavy particles toward the core for $D = 2.5$ and 3.1 were exceptionally rapid, being very similar to results obtained in [2], while the resulting motion for $D = 2.3$ were relatively slow. All the above considerations led to the parameter choices $\dot{\theta} = 4$, $\varepsilon = 10$, $D = 2.3$ and $k^* = 50$.

The initial positions, velocities, and temperatures for the first two examples to be described next were generated in the manner prescribed in Section 3 and, for completeness, are given in Table I.

Example 1. In this example, let $G^*_{ij} \equiv 0$. Figure 2 shows the initial particle configuration given in Table 1 at time t_0 . The relative mass of each particle is indicated by the relative size of the circle which represents that particle. For emphasis, only the smallest circles, which represent masses of 2000 units, are unshaded. Figures 3-5 show the evolution of the system at times t_{5500}, t_{18500} and t_{38500} . By t_{5500} , two particles have escaped from the system; by t_{18500} , forty four have escaped; by t_{38500} , eighty seven have escaped. Thereafter, no additional particles leave the system. Figures 3-5 reveal a gradual restratification of the particles with the heavier ones relocating centrally and the lighter ones to the outside. Noting that this effect results in the absence of gravitation, since $G^*_{ij} \equiv 0$, leads to the deduction that the short range forces do have important global effects. Figure 6 shows a typical set of escaping particles at t_{18500} and, here, too, the heaviest particles are at the core of this subsystem.

Example 2. Let all the parameters and initial data be the same as in Example 1 with the single exception that $G^*_{ij} \equiv 10^{-3}$, thus introducing self-gravitation. Figure 7 shows the system at t_{5500} and reveals a greater cohesiveness than that shown in Figure 3. Figure 8(b) shows

Figure 2 - THE INITIAL CONFIGURATION



Figure 3 - $T = t_{5500}$ (Example 1)



Figure 4 - $T = t_{18500}$ (Example 1)



Figure 5 - $T = t_{38500}$ (Example 1)



Figure 6 - AN ESCAPING SUBSYSTEM (Example 1)



Figure 7 - $T = t_{5500}$ (Example 2)



the system at time t_{11000} and allows a direct comparison with the case $G^*_{ij} \equiv 0$, shown in (a). Again, the greater cohesiveness is easily apparent. By the time t_{15000}, many of the light, outer particles have begun to leave the system, but have been pulled back quickly by gravitation. Finally, at t_{18000} even the present system subdivides, but this time it is, essentially, into two relatively large subunits, as shown in Figure 9.

Before proceeding to Example 3, note that Examples 1 and 2, above, demonstrate quite clearly the well known fact that heavy particles and systems of such particles on the periphery of a rotating fluid system will often have sufficient momentum to escape the system. Since this is always valid, we will turn next to an example in which relatively heavy particles are not located near the system's outer boundary. That is, we will start with a system from which some of the particles which will escape are considered to have already escaped.

Example 3. In this example, consider all the parameters to be the same as those in Example 2 except that the masses of thirteen pairs of particles will be interchanged in the fashion shown in Figure 10. The resulting mass distribution at time t_0 is shown in Figure 11. The initial spin up motion is shown at t_{5500} and t_{11500} in Figures 12 and 13. The velocity fields at these times are given in Figures 14 and 15. The motion of the system from t_{18500} to t_{171500} is shown in Figures 16(a)-(i). By the time t_{18500}, a large concentration of the heavier particles has already formed, as is seen in Quadrant II of Figure 16(a).

Figure 8 - $T = t_{11000}$ (Examples 1 and 2)



Figure 9 - $T = t_{18500}$ (Example 2)



NO. 341-10 DIETZGEN GRAPH PAPER 10 X 10 PER INCH EUGENE DIETZGEN CO. MADE IN U. S. A.

Figure 10 - MASS REDISTRIBUTION



Figure 11 - MASS DISTRIBUTION AT t_0 (Example 3)



Figure 12 - $T = t_{5500}$ (Example 3)



Figure 13 - $T = t_{11500}$ (Example 3)



Figure 14 - VELOCITY FIELD AT t_{5500}



Figure 15 - VELOCITY FIELD AT t_{11500}



Figure 16(a) - $T = t_{18500}$ (Example 3)



Figure 16(b) - $T = t_{27500}$ (Example 3)



Figure 16(c) - T = t_{38500} (Example 3)



Figure 16(d) - T = t_{51500} (Example 3)



Figure 16(e) - $T = t_{66500}$ (Example 3)



Figure 16(f) - $T = t_{81500}$ (Example 3)



Figure 16(g) - $T = t_{111500}$ (Example 3)



Figure 16(h) - $T = t_{141500}$ (Example 3)



Figure 16(i) - T = t_{171500} (Example 3)



Figure 17(a) - T = t_{18500} (Example 3)



Note, incidentally, that one of the lightest particles has been captured within this heavy concentration and that it never escapes. By the time t_{27500} , as shown in Figure 16(b), a second smaller concentration of the heavier particles, each with mass 4000 units, has formed. Also, only particles of mass 2000 units now form an outer boundary of the system. Figures 16(b)-(f) reveal the gravitational motion of the two relatively heavy mass concentrations towards each other, but these never do join, as will be analyzed later. The ease with which the boundary of the system changes shape, as evidenced in Figures 16(c)-(i), clearly indicates the fluid nature of the system, but, the onset of solidification is suggested by the general elliptic shape of the system at t_{111500} and t_{141500} , and the almost circular shape at t_{171500} . The process of solidification is shown in Figures 17(a)-(i), in which the system is again shown at the exact time steps as those of Figures 16(a)-(i), but with the physical state of each particle, as determined from Table IV, indicated in the following fashion. Each particle which is solid has been enclosed by a hexagon. Each particle which is gaseous has been enclosed by a triangle. All other particles are liquid. At t_{18500} one sees in Figure 17(a) that the particles of mass 6000, 8000, and 10000 are all solid, that particles of mass 4000 are about half solid and half liquid, and that most of the particles of mass 2000 are liquid, with some few being solid and some few being gaseous. This preponderance of liquid particles prevails through t_{81500} and accounts for the highly fluid motion of the system up to this time. However, at t_{81500} , one sees that solidification has set into the area between the two relatively heavy mass centers, which, of course, accounts for the cessation of their motion towards each other. This increase of solidification is seen in Figures 17(g)-(i) and accounts for a highly fluid behavior at the outer boundaries of the system, but not in the interior. At t_{171500} , as seen in Figure 17(i), large sections of the boundary seem to have solidified, thus suggesting the beginning of crustal formation. Unfortunately, the contraction of the system and the decrease in system angular velocity by this time resulted in such a slow net solidification rate that the amount of computation required to continue to complete solidification became economically unfeasible. This decrease in system angular velocity is shown in Figure 18, where each unit on the horizontal axis corresponds to 10000 times steps. The oscillatory nature of the decay is clear from the figure. Note also, that whereas the system completed a revolution in about 15000 times steps at the earlier times, it required about 34000 time steps at the later times. It was for this reason that the time steps shown in Figures 16(a)-(i) were staggered in the indicated fashion.

It is of special importance to note that our definition and treatment of solidification actually corresponds to rigid motions of sets of solid particles.

Figure 17(b) - T = t_{27500} (Example 3)



Figure 17(c) - T = t_{38500} (Example 3)



Figure 17(d) - T = t_{51500} (Example 3)



Figure 17(e) - T = t_{66500} (Example 3)





Figure 17(f) - $T = t_{81500}$ (Example 3)



Figure 17(g) - $T = t_{111500}$ (Example 3)

Figure 17(h) - T = t_{141500} (Example 3)



Figure 17(i) - T = t_{171500} (Example 3)



Figure 18 - SYSTEM ANGULAR VELOCITY



5. Conclusions and Remarks

Example 3 of Section 4 is indicative of the large variety of results which can be obtained simply by varying input parameters and/or initial data. With regard to Example 3, in particular, we hypothesize that it approximates the development of a moon-like body, primarily because, like the Earth's moon, its geometric center and its mass center are distinctly different. Indeed, a subsystem which escaped at the time when the earth had cooled to a molten state might reasonably have the initial data prescribed in Example 3. We hypothesize further that to simulate the development of Earth, itself, Example 3 should be rerun with the following necessary modifications: (a) use masses 4000, 3500, 3000, 2500, 2000 instead of 10000, 8000, 6000, 4000, 2000, respectively, to ensure that the initial particle states are all either gas or liquid; (b) use a "shadow-region" approach [2] to define the system's boundary at each time step and involve all and only boundary particles in radiative heat transfer; and (c) increase the damping factor in (3.13) so that solidification occurs more slowly, thus allowing the formation of a circular core. Should these modifications prove to be relatively successful, then further refinement could be obtained by introducing chemical reactions into the model.

References

1. J. A. Barker and D. Hendersen, "What is 'liquid'? Understanding the states of matter", Rev. Mod. Phys., 48, 1976, pp 587-671.

2. D. Greenspan, Discrete Models, Addison-Wesley, Reading, Mass., 1973.

3. __________, "Computer studies of interactions of particles with differing masses", TR 285, Comp. Sci. Dept., U. of Wis., Madison, 1977, to appear in Journal of Computational and Applied Mathematics.

4. D. Greenspan and J. Collier, "Computer studies of swirling particle fluids and the evolution of planetary-type bodies", TR 299, Comp. Sci. Dept., U. of Wis., Madison, 1977.

5. J. O. Hirschfelder, C. F. Curtiss and R. B. Bird, Molecular Theory of Gases and Liquids, Wiley, N.Y., 1954.

APPENDIX

FORTRAN PROGRAM - CREATE

```
C FORTRAN PROGRAM FOR THE CREATION OF THE EARTH.
C RESULTS TO BE PUBLISHED IN OSSERVATORE ROMANO, SUNDAY EDITION.
      PARAMETER N=239, N2=N*2
C DIMENSION STATEMENT
      DIMENSION PMASS(N),XO(N),YO(N),VXO(N),VYO(N),X(N,2),Y(N,2),
     1VX(N,2),VY(N,2),ACX(N),ACY(N),TEMP(N),XREL(N),YREL(N),DIST(N),
     1VTREL(N),VNREL(N),VXREL(N),VYREL(N)
C  VARIABLES XO, YO, ETC., HAVE BEEN PUNCHED OH, NOT ZERO
C INITIALIZATION SET TO ZERO AUTOMATICALLY ON 1110
C WE NEXT ASSURE THAT X, Y, VX, VY, ACX, ACY ARE JUXTAPOSED WITHIN CORE.
      COMMON/DG/X,Y,VX,VY,ACX,ACY
      NM1=N-1
      K=1
      KDAMP=50
      KPRINT=100
C INITIAL DATA INPUT
      READ 10, (PMASS(I),XO(I),YO(I),VXO(I),VYO(I),TEMP(I),I=1,N)
10      FORMAT (5F10.3,F10.1)
C PRINT INITIAL DATA
      DO 20 I=1,N
      PRINT 15,I,PMASS(I),XO(I),YO(I),VXO(I),VYO(I),TEMP(I)
15       FORMAT (5X,I7,5F12.5,F12.1)
20       CONTINUE
C UPDATE. X(I,1) IS X COORDINATE AT PREVIOUS TIME STEP.  X(I,2) IS X
C  COORDINATE AT PRESENT TIME STEP.  SIMILARLY FOR OTHER VARIABLES.
C INITIALLY WE WOULD LIKE TO DO THE FOLLOWING FOR I=1,N
C      X(I,1)=XO(I)
C      Y(I,1)=YO(I)
C      VX(I,1)=VXO(I)
C      VY(I,1)=VYO(I)
C INSTEAD WE DO IT BY THE SUBROUTINE AS FOLLOWS.
      CALL COPVAR (XO(1),1,X(1,1),1,N)
      CALL COPVAR (YO(1),1,Y(1,1),1,N)
      CALL COPVAR (VXO(1),1,VX(1,1),1,N)
      CALL COPVAR (VYO(1),1,VY(1,1),1,N)
       GO TO 71
C WE WILL WANT TO UPDATE
C      X(I,1)=X(I,2)
C      Y(I,1)=Y(I,2)
C      VX(I,1)=VX(I,2)
C      VY(I,1)=VY(I,2)
C WHICH IS ACCOMPLISHED BY
65      CALL COPVAR (X(1,2),1,X(1,1),1,N)
      CALL COPVAR (Y(1,2),1,Y(1,1),1,N)
      CALL COPVAR (VX(1,2),1,VX(1,1),1,N)
      CALL COPVAR (VY(1,2),1,VY(1,1),1,N)
C CALCULATION OF ACCELERATIONS IS DONE THROUGH STEP 770
C      DO 701 I=1,N
C      ACX(I)=0.
C      ACY(I)=0.
C 701      CONTINUE
```

```
C ARE ACCOMPLISHED BY
      CALL COPCON (0.,0,ACX(1),1,N)
      CALL COPCON (0.,0,ACY(1),1,N)
71     DO 77 I=1,NM1
      IP1=I+1
      DO 76 J=IP1,N
      R2=(X(I,1)-X(J,1))**2+(Y(I,1)-Y(J,1))**2
      R=SQRT(R2)
      IF (R2.GT.5.29) GO TO 72
      F=5.*PMASS(I)*PMASS(J)/(R**6)-5.*PMASS(I)*PMASS(J)/(R**4)
      FX=F*(X(I,1)-X(J,1))/R
      FY=F*(Y(I,1)-Y(J,1))/R
      GO TO 75
72       F=-0.001*PMASS(I)*PMASS(J)/(R2)
      FX=F*(X(I,1)-X(J,1))/R
      FY=F*(Y(I,1)-Y(J,1))/R
C ACCUMULATION OF FORCES ON PARTICLE I DUE TO ALL OTHER PARTICLES IS DONE
C IN NEXT FOUR FORMULAS
75     ACX(I)=ACX(I)+FX
      ACX(J)=ACX(J)-FX
      ACY(I)=ACY(I)+FY
      ACY(J)=ACY(J)-FY
76     CONTINUE
77     CONTINUE
C NOTE THAT WE HAVE JUST ACCUMULATED FORCES, NOT ACCELERATIONS - THE ABOVE
C NOTATION, THOUGH MISLEADING, ENABLES US TO SAVE MEMORY LOCATIONS.  WE
C NEXT CALCULATE THE VELOCITIES AND POSITIONS AS FOLLOWS.
C ACTUAL ACCELERATIONS ARE INCLUDED BY DIVISION BY MASS OF 75 DIRECTLY.
C LEAP FROG FORMULAS ARE USED BUT SPECIAL STARTERS ARE NOT SHOWN BECAUSE
C PROGRAM HERE CONTINUES CALCULATIONS FROM PUNCHED OUTPUT.
79     DO 799 I=1,N
      VX(I,2)=VX(I,1)+.0001*ACX(I)/PMASS(I)
      VY(I,2)=VY(I,1)+.0001*ACY(I)/PMASS(I)
      X(I,2)=X(I,1)+.0001*VX(I,2)
      Y(I,2)=Y(I,1)+.0001*VY(I,2)
799     CONTINUE
800     K=K+1
C CALCULATE MASS CENTER AND AVERAGE VELOCITY
      XBAR=0.
      YBAR=0.
      VXBAR=0.
      VYBAR=0.
      DO 8003 I=1,N
      XBAR=XBAR+PMASS(I)*X(I,2)
      YBAR=YBAR+PMASS(I)*Y(I,2)
      VXBAR=VXBAR+PMASS(I)*VX(I,2)
      VYBAR=VYBAR+PMASS(I)*VY(I,2)
8003    CONTINUE
      XBAR=XBAR/618000.
      YBAR=YBAR/618000.
      VXBAR=VXBAR/618000.
      VYBAR=VYBAR/618000.
      PRINT 8006,XBAR,YBAR,VXBAR,VYBAR
8006      FORMAT (5X,4F10.2)
```

```
C FINALLY WE CALCULATE TEMPERATURE. THIS IS DONE RELATIVE TO (XBAR,YBAR).
C WE TAKE OUT ALL LINEAR AND ROTATIONAL SYSTEM VELOCITIES FROM EACH PARTICLE.
C WHAT REMAINS IS THE INTERPARTICLE MOTIONS, WHICH DEFINE THE PARTICLES HEAT.
C WE FIRST SHIFT TO THE CENTER OF MASS.
      DO 8020 I=1,N
      XREL(I)=X(I,2)-XBAR
      YREL(I)=Y(I,2)-YBAR
      VXREL(I)=VX(I,2)-VXBAR
      VYREL(I)=VY(I,2)-VYBAR
8020   CONTINUE
C WE NEXT INTRODUCE TANG AND NORM VELOCITY COMPONENTS.
      DO 8022 I=1,N
      DIST(I)=SQRT(XREL(I)**2+YREL(I)**2)
      VTREL(I)=(-VXREL(I)*YREL(I)+VYREL(I)*XREL(I))/DIST(I)
      VNREL(I)=(VYREL(I)*YREL(I)+VXREL(I)*XREL(I))/DIST(I)
8022   CONTINUE
C OUT OF VTREL(I) TAKE THE AVERAGE ANGULAR VELOCITY OF THE SYSTEM.
C THIS AVERAGE ANGULAR VELOCITY IS THDOTB.
      THDOTB=0.
      DO 8024 I=1,N
      THDOTB=THDOTB+PMASS(I)*VTREL(I)/DIST(I)
8024   CONTINUE
      THDOTB=THDOTB/548000.
      PRINT 8026,THDOTB
8026   FORMAT (5X,F10.5)
C WE NOW SUBTRACT OFF THE SYSTEM ROTATIONAL VELOCITY.
      DO 8028 I=1,N
      VTREL(I)=VTREL(I)-DIST(I)*THDOTB
8028   CONTINUE
C FINALLY WE CALCULATE TEMP.  IT IS NORMALIZED BY DIVISION BY 1000 AND
C IS TAKEN AS THE AVERAGE OVER 500 TIME STEPS TO INDICATE THAT IT IS AN
C OBSERVED QUANTITY.  IT IS ESSENTIALLY KE.
      DO 8029 I=1,N
      TEM=VTREL(I)**2+VNREL(I)**2
      TEM=.0005*PMASS(I)*TEM
       TEMP(I)=(499.*TEMP(I)+TEM)/500.
8029      CONTINUE
C RADIATE , IN AND OUT, EVERY KDAMP STEPS.
      IF (MOD(K,KDAMP).GT.0) GO TO 8030
C WE TAKE ACCOUNT OF RADIATION FOR ALL OUTER PARTICLES, WHICH MEANS
C ALL PARTICLES A DISTANCE GREATER THAN 7 FROM THE CENTROID, BY ADDING
C OR SUBTRACTING FROM THEIR SPEEDS.  WE WORK WITH SQUARES OF SPEEDS IN
C ORDER TO ECONOMIZE ON THE SQUARE ROOT PROCEDURE.
C ASSUME THE SUN IS ABOVE THE CENTROID AND DARKNESS BELOW.
      DO 8018 I=1,N
       DISTM=(X(I,2)-XBAR)**2+(Y(I,2)-YBAR)**2
       IF (DISTM.LT.49.) GO TO 8018
      IF (Y(I,2).GT.YBAR) GO TO 8012
      VX(I,2)=0.9955*VX(I,2)
      VY(I,2)=0.9955*VY(I,2)
      GO TO 8018
8012      VX(I,2)=1.001*VX(I,2)
      VY(I,2)=1.001*VY(I,2)
8018      CONTINUE
```

```
8030    IF (MOD(K,KPRINT).GT.0) GO TO 82
      DO 810 I=1,N
       PRINT 81,K,I,X(I,2),Y(I,2),VX(I,2),VY(I,2),TEMP(I)
81      FORMAT (5X,2I7,5F12.4)
810      CONTINUE
C TERMINATION AFTER A FIXED NUMBER OF STEPS
82      IF (K.LT.500) GO TO 65
C PUNCH OUTPUT FOR RESTART
      PUNCH 10, (PMASS(I),X(I,2),Y(I,2),VX(I,2),VY(I,2),TEMP(I),I=1,N)
      STOP
      END
```